פלט פשוט
ל-awk  יש פקודה  print  לבצע הדפסות פשוטות.  ביצוע הפקודה הזו מדפיסה שורת פלט אחת ובאופן עקרוני משמשת להדפסת תוכני משתנים בהפרש של מספר רווחים זה מזה.
  פורמט הפקודה הזו היא 

print  רשימת משתנים מופרדים ע"י פסיקים
רשימת המשתנים היא אופציונלית.  ביצוע הפקודה ללא פרמטרים תדפיס את "השורה הנוכחית" מהקובץ  (באופן שקול: print $0).  
לדוגמא

print $1,  $2*$3

פירושו הדפסת תוכן השדה הראשון ומכפלת השני והשלישי של השורה הנוכחית.

הפקודה 

print $1, $NF

ידפיס את השדה הראשון והאחרון בשורה הנוכחית.  NF הינו משתנה מובנה של השפה המכיל את מספר השדות בשורה הנוכחית
הפקודה

print NR, $0

ידפיס את השורה הנוכחית יחד המספר הסידורי שלו בקובץ.  NR הוא משתנה מובנה בשפת awk  כמה שורות מהקובץ קראנו עד עכשיו.  
 אפשר גם להדפיס מחרוזות ביחד עם תוכני משתנים:

print " x = ", x
הפקודה print ניתן לכתוב גם בצורת סוגריים למשל

print("x =", x)

פלט מובנה
   בשפת awk  ישנה פקודת printf  הדומה מאד לפקודה המקבילה בשפת C.
את התוכנית לחישוב סכומים לתשלום אפשר להמיר ל-
$3 > 0 {printf("Total pay for %s is %.2f\n", $1, $2 * $3)}
ואז הפלט יהיה

Total pay for Gadi is 40.00
Total pay for Haim is 100.00
Total pay for Maira is 121.00
Total pay for Yossi is 76.50
בחירה על פי תנאי
  ביצוע של הפעולה שבין הסוגריים המסולסלות {} יכול להיות מותנה בתנאי לוגי כפי שהדבר נראה בדוגמא האחרונה.  התנאי יכול להיות מספרי כמו

$3 >= 5

$2 * $3 > $1

טקסטואלי כמו

$2 == "Hello"

ומורכב לוגית

($2 * $3 > $1) && ($2 == "Hello")

הנחיות BEGIN ו-END
ל-awk יש אמצעים לביצוע קוד לפני הסריקה של שורות קובץ הטקסט ואחריו.  ההנחיות הללו הם BEGIN ו-END בהתאמה.
לדוגמא נניח שנרצה להדפיס תוכן של קובץ השעות עם כותרת מקדימה והודעת סיום בסוף ההדפסה.
התוכנית תהיה

BEGIN { print "NAME  RATE HOURS"; print ""}

      { print }
END {print "END OF FILE"}
והפלט יהיה

NAME  RATE HOURS

Benny 4.00 0
Dorit  3.75 0
Gadi 4.00 10
Haim 5.00 20
Maira 5.50 22
Yossi 4.25 18
END OF FILE
ביצוע חישובים בעזרת awk
שפת awk  תומכת במבני התכנות הרגילים כולל שימוש במשתנים, לולאות וקריאה לפונקציות.   אבל יש מספר הבדלים לעומת שפות תכנות "רגילות" כמו C או פסקל.  
· קודם כל ההקשר של החישוב הוא במסגרת סריקה של קובץ או קובצי טקסט.  
· מעבר לכך, ההנחה היא שמדובר בעיקר בעיבוד של מחרוזות או מספרים.  מסיבה זו משתנים ב-awk הם באופן מוזר יכולים להיות גם מחרוזות וגם מספרים, אין צורך להכריז עליהם ויש פונקציות מובנות המאפשרות עיבוד נוח של המשתנים בהקשר המאד מסוים הזה.  משתנים ב-awk מאותחלים באופן אוטומטי באפס אם ההתייחסות אליו הוא מספרי או מחרוזת ריקה אם ההתייחסות עליו כמחרוזת.
לדוגמא, נניח שאנחנו רוצים לחשב את סכום הכולל של השכר בקובץ העובדים – תעריף – שעות אזי התוכנית הבאה תעשה זאת
# pay.awk
{ pay = pay + $2 * $3}
END {  print NR, "employees"
             print "total pay is", pay
             print "average pay is ", pay/NR
    }
 ואם נריץ את התוכנית ע"י הפקודה 

awk –f pay.awk emp.dat

נקבל פלט
7 employees

total pay is 337.5
average pay is  48.2143
טיפ: הדפסת השורה האחרונה
  למרות ש-NR שומר על ערכו בסוף הקלט זה לא נכון עבור 0$.  לפיכך אם אנחנו רוצים לעשות משהו עם השורה האחרונה (להדפיס אותו למשל) צריך לשמור אותו במשתנה.  למשל התוכנית הבאה מדפיסה את השורה האחרונה:
{last = $0}

END { print last }
פונקציות מובנות
מלבד פונקציות מובנות אריתמטיות awk   עשירה בפונקציות מובנות למניפולת\ציה של טקסט.  אחד מהם הוא length  המתפקד בצורה דומה ל-strlen של  שפת C.
 לדוגמא, התוכנית הבאה סופרת את מספר התווים, שורות ומילים בקובץ טקסט.

# wc.awk - count characters, words, line

  { nc = nc + length($0) +1
    nw = nw + NF
  }
END { print NR, "lines", nw, "words" , nc, "characters" }
הפלט של הריצה של הפקודה

awk -f wc.awk emp.dat

תהיה:

6 lines 18 words 81 characters
תכנות מבני
  שפת awk  תומכת בקוד מבני if, else, while  ו-for בגרסאות דומות לאלו בשפת C. ההקשר של האמצעים הללו בדרך כלל הוא להריץ קטע קוד מורכב על כל שורה בקובץ או קובצי הקלט, או לפני קריאת הקבצים או בסופם. 
  במידה ותחת מבנה אמור להתבצע פקודה מורכבת הם מוכללות בסוגריים מסולסלות {}.
דוגמא ל-if – else : תוכנית הסורקת את קובץ השעות וההדפסה בסופה תהיה מותנה בכך שיש עובדים שיש להם שעות עבודה בחודש:

# ifelse.awk
 $3 > 0 { n = n +1; pay = pay + $2 * $3  }
  END {
       if (n > 0)
          print n, " employees, total pay is", pay,
                   "avarage pay is", pay/n
       else
         print "No employees worked hours this month"
      }
דוגמא ל-while:
  נניח שיש לנו קובץ של סכומים, אחוז ריבית ומספר שנים, ואנחנו רוצים לחשב את הריבית דריבית לאורך השנים של הסכום. התוכנית:
# interest1.awk - compute compound interest

# input : amount rate years
# output: compounded value at the end of each year
{ i = 1
  print("amount = ", $1, "rate = ", $2, "years = ", $3)
  while (i <= $3) {
                  printf("\t%.2f\n", $1 * (1+$2)^ i)
                  i = i + 1
                  }
}
אם תוכן קובץ הקלט הוא

1000 0.035 5
1000 0.055 5
1000 0.024 5
אזי הפלט יהיה

amount =  1000 rate =  0.035 years =  5

        1035.00
        1071.22
        1108.72
        1147.52
        1187.69
amount =  1000 rate =  0.055 years =  5
        1055.00
        1113.02
        1174.24
        1238.82
        1306.96
amount =  1000 rate =  0.024 years =  5
        1024.00
        1048.58
        1073.74
        1099.51
        1125.90
מימוש אותו רעיון בעזרת for:

# interest2.awk - compute compound interest

# input : amount rate years
# output: compounded value at the end of each year
{
  print("amount = ", $1, "rate = ", $2, "years = ", $3)
  for (i=1; i <= $3; i++)
                  printf("\t%.2f\n", $1 * (1+$2)^ i)
}
מערכים
שפת awk  תומכת במערכי משתנים.  כרגיל אין הכרזות  ב-awk, גם לא על מערכים ויש אתחול אוטומטי.  בניגוד לשפת C האינדקס של האיבר הראשון במערך הוא 1 (ולא אפס).  גודל המערך נקבע לפי האינדקס המרבי שניגשים אליו. 
  לדוגמא, תוכנית המדפיסה את קובץ הקלט בסדר הפוך:

# reverse.awk print input file in reverse order
{x[NR]= $0}
END{ for (i=NR; i > 0; i--)print x[i]}
הפלט במקרה של ריצה על הקובץ emp.dat:

Yossi 4.25 18
Maira 5.50 22
Haim 5.00 20
Gadi 4.00 10
Dorit  3.75 0
Benny 4.00 0
